
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test cases 7.1.2.2.5 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2018-12_D19wk23’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.2.5
Test Group:
\EN-DC\ 7_1_2\
ATS Version:
iwd-TTCN3-B2018-12_D19wk23
System Simulator used:
Starpoint SP9500 Test system

UE used:
HiSilicon Balong 5000&MediaTek MT6297
Verification Status:
PASS


4. Corrections required for ENDC test case 7.1.2.2.5 using wk23
Change 4.1
	Function name
	f_TC_7_1_2_2_5_NR_TestBody

	Reason for change
	1. Test case doesn’t require UE loop back RLC SDU in segments, and for the first UL data, 2 octets BSR should be taken into consideration. Therefore the TBS for UL resource allocation in current TTCN is not enough for the complete RLC SDU.
2. At step 3A and step 3B, when RLC SDU #5 with SN=w-1 was sent, the SN of RLC SDU #1 just falls on the edge of the reassembly window according to TS 38.322 cl.5.2.2.2.1:
The receiving UM RLC entity shall maintain a reassembly window according to state variable RX_Next_Highest as follows:

-
a SN falls within the reassembly window if (RX_Next_Highest – UM_Window_Size) <= SN <RX_Next_Highest;

-
a SN falls outside of the reassembly window otherwise.

Therefore UE will loopback the RLC SDU #1 afterwards and TP1 will not be achieved.

3. The UL PDCP count is maintained by UE, the altstep for receiving UM PDU should not use DL PDCP PDU as the parameter in receiving template.
 

	Summary of change
	1. Change TBS from 80 to 144(12 octets RLC SDU + 1 octet RLC header + 2 octets MAC header + 2 octets BSR = 17octets/136 bits).
2. Change SN from w-1 to w, so that the SN of RLC SDU #1 can falls outside of the reassembly window.
3. Reset the NR_RLC_Data with the right UL PDCP SN before receiving RLC SDU(step4, step4A, step13).

	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	


Before change

  function f_TC_7_1_2_2_5_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                            




  DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { //UM RLC / Receive window operation and t-Reassembly expiry

    var integer v_NR_UM_Window_Size;

    var integer v_RLC_Len1stSeg := 7; // Length of 1st segment

    var integer v_RLC_Len2ndSeg := 5; // Length of 2nd segment

    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg;

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var SubFrameTiming_Type v_TimingDL4;

    var SubFrameTiming_Type v_TimingDL5;

    var SubFrameTiming_Type v_TimingDL6;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var float v_T_ReassemblyMax := f_NR_SetTimerToleranceMax (nr_Cell1, l2Timer, p_RLC_Rec.t_Reassembly);

    timer t_WatchDog;

    if (p_RLC_Rec.UM_SN_Size == 6) {

      v_NR_UM_Window_Size := tsc_NR_UM_Window_Size_6;

      // TBS for the first segment = RLC PDU data + RLC header without SO(1 octet) + MAC header(2 octets) = 7 + 1 + 2 = 10 octets => TBS = 80 bits
      // TBS for the last segment = RLC PDU data + RLC header with SO(2 octets) + MAC header(2 octets) = 5 + 3 + 2 = 10 octets => TBS = 80 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(80, v_NR_UplinkBWP);

    } else {

      v_NR_UM_Window_Size := tsc_NR_UM_Window_Size_12;

      // TBS for the first segment = RLC PDU data + RLC header without SO + MAC header = 7 + 2 + 2 = 11 octets => TBS = 88 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO + MAC header = 5 + 4 + 2 = 11 octets => TBS = 88 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);

    }

    fl_NR_GenerateRLC_UM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 5);

    // @siclog "Step 0" siclog@

    // The SS ignores scheduling requests and does not allocate any uplink grant.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#1

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 40);

    v_TimingDL3 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 60);

    v_TimingDL4 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 80);

    v_TimingDL5 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 100);

    v_TimingDL6 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 160); // after the waiting time for the possible response, but well before 200 ms (t-Reassembly)

    // @siclog "Step 1" siclog@

    //The SS transmits UMD PDU#1 containing first segment of RLC SDU#1, SN=0.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU1, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 2" siclog@

    //20 ms after step 1 the SS transmits UMD PDU#3 containing first segment of RLC SDU#2, SN=1.

    p_RLC_Rec.UM_TX_Next := 1;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU2, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 3" siclog@

    //40 ms after step 1 the SS transmits UMD PDU#4 containing last segment of RLC SDU#2, SN=1.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL2), tsc_RLC_SDU2, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 3A" siclog@

    //60 ms after step 1 the SS transmits UMD PDU#9 containing first segment of RLC SDU#5, SN=w-1.

    p_RLC_Rec.UM_TX_Next := v_NR_UM_Window_Size-1;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL3), tsc_RLC_SDU5, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 3B" siclog@

    //80 ms after step 1 the SS transmits UMD PDU#10 containing last segment of RLC SDU#5, SN=w-1.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL4), tsc_RLC_SDU5, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    //@siclog "Step 3C" siclog

    // 100 ms after step 1 the SS assigns 2 UL grants (UL grant allocation type 2) with a time spacing of 20 ms so as to loop back RLC SDU#2 and RLC SDU#5.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL5, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 5" siclog@

    //160 ms after step 1 the SS transmits UMD PDU#2 last segment of RLC SDU#1, SN=0.

    p_RLC_Rec.UM_TX_Next := 0;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL6), tsc_RLC_SDU1, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 4" siclog@

    //Check: Does the UE transmit RLC SDU#2?

    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 4A" siclog@

    //Check: Does the UE transmit RLC SDU#5?

    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU5);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4A");

    // @siclog "Step 6" siclog@

    //Check: For 1 sec after step 5, does the UE transmit RLC SDU#1?

    t_WatchDog.start(1.16); // as step 5 is scheduled: 1.0 sec + 160 msec

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");}

      [] t_WatchDog.timeout {}

    }

    // @siclog "Step 6A" siclog

    //The SS starts the UL default grant transmissions

    f_NR_ULGrantConfiguration_Start(nr_Cell1);

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#3

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 240); // > v_T_Reassembly !!

    // @siclog "Step 7" siclog@

    //The SS transmits UMD PDU#5 containing first segment of RLC SDU#3, SN=5.

    p_RLC_Rec.UM_TX_Next := 5;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 8" siclog@

    //Wait for 200 ms to ensure that t-Reassembly for the UMD PDU#5 expires.

    f_Delay(v_T_ReassemblyMax);

    // @siclog "Step 9" siclog@

    //The SS transmits UMD PDU#6 containing last segment of RLC SDU#3, SN=5.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU3, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 10" siclog@

    //Check: For 1 sec after step 9, does the UE transmit RLC SDU#3?

    t_WatchDog.start(1.24); // as step 9 is scheduled: 1.0 sec + 240 msec

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");}

      [] t_WatchDog.timeout {}

    }

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#4

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    // @siclog "Step 11" siclog@

    //The SS transmits UMD PDU#7 containing first segment of RLC SDU#4, SN=8.

    p_RLC_Rec.UM_TX_Next := 8;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 12" siclog@

    //The SS transmits UMD PDU#8 containing last segment of RLC SDU#4, SN=8.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 13" siclog@

    //Check: Does the UE transmit RLC SDU#4?

    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");

  }

After change

    function f_TC_7_1_2_2_5_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                            




     DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { //UM RLC / Receive window operation and t-Reassembly expiry

    var integer v_NR_UM_Window_Size;

    var integer v_RLC_Len1stSeg := 7; // Length of 1st segment

    var integer v_RLC_Len2ndSeg := 5; // Length of 2nd segment

    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg;

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var SubFrameTiming_Type v_TimingDL4;

    var SubFrameTiming_Type v_TimingDL5;

    var SubFrameTiming_Type v_TimingDL6;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var float v_T_ReassemblyMax := f_NR_SetTimerToleranceMax (nr_Cell1, l2Timer, p_RLC_Rec.t_Reassembly);

    timer t_WatchDog;

    if (p_RLC_Rec.UM_SN_Size == 6) {

      v_NR_UM_Window_Size := tsc_NR_UM_Window_Size_6;

      // TBS for UL data  = RLC PDU data + RLC header without SO(1 octet) + MAC header(2 octets) + BSR(2 octets)= 12 + 1 + 2 + 2= 17 octets => TBS = 144 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(144, v_NR_UplinkBWP);

    } else {

      v_NR_UM_Window_Size := tsc_NR_UM_Window_Size_12;

      // TBS for UL data = RLC PDU data + RLC header without SO + MAC header  + BSR(2 octets) = 12 + 2 + 2 +2 = 18 octets => TBS = 144 bits

      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(144, v_NR_UplinkBWP);//12+1+3+2(BSR)

    }

    fl_NR_GenerateRLC_UM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 5);

    // @siclog "Step 0" siclog@

    // The SS ignores scheduling requests and does not allocate any uplink grant.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#1

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 40);

    v_TimingDL3 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 60);

    v_TimingDL4 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 80);

    v_TimingDL5 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 100);

    v_TimingDL6 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 160); // after the waiting time for the possible response, but well before 200 ms (t-Reassembly)

    // @siclog "Step 1" siclog@

    //The SS transmits UMD PDU#1 containing first segment of RLC SDU#1, SN=0.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU1, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 2" siclog@

    //20 ms after step 1 the SS transmits UMD PDU#3 containing first segment of RLC SDU#2, SN=1.

    p_RLC_Rec.UM_TX_Next := 1;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU2, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 3" siclog@

    //40 ms after step 1 the SS transmits UMD PDU#4 containing last segment of RLC SDU#2, SN=1.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL2), tsc_RLC_SDU2, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 3A" siclog@

    //60 ms after step 1 the SS transmits UMD PDU#9 containing first segment of RLC SDU#5, SN=w.

    p_RLC_Rec.UM_TX_Next := v_NR_UM_Window_Size;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL3), tsc_RLC_SDU5, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 3B" siclog@

    //80 ms after step 1 the SS transmits UMD PDU#10 containing last segment of RLC SDU#5, SN=w.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL4), tsc_RLC_SDU5, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    //@siclog "Step 3C" siclog

    // 100 ms after step 1 the SS assigns 2 UL grants (UL grant allocation type 2) with a time spacing of 20 ms so as to loop back RLC SDU#2 and RLC SDU#5.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL5, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);

    // @siclog "Step 5" siclog@

    //160 ms after step 1 the SS transmits UMD PDU#2 last segment of RLC SDU#1, SN=0.

    p_RLC_Rec.UM_TX_Next := 0;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL6), tsc_RLC_SDU1, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 4" siclog@

    //Check: Does the UE transmit RLC SDU#2?

p_RLC_Rec.NR_RLC_DataList[1] := '8000005408ABEECE4B0B813F'O;
    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // @siclog "Step 4A" siclog@

    //Check: Does the UE transmit RLC SDU#5?

p_RLC_Rec.NR_RLC_DataList[4] := '80000195688EBE465F09816A'O;
    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU5);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4A");

    // @siclog "Step 6" siclog@

    //Check: For 1 sec after step 5, does the UE transmit RLC SDU#1?

    t_WatchDog.start(1.16); // as step 5 is scheduled: 1.0 sec + 160 msec

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");}

      [] t_WatchDog.timeout {}

    }

    // @siclog "Step 6A" siclog

    //The SS starts the UL default grant transmissions

    f_NR_ULGrantConfiguration_Start(nr_Cell1);

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#3

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 240); // > v_T_Reassembly !!

    // @siclog "Step 7" siclog@

    //The SS transmits UMD PDU#5 containing first segment of RLC SDU#3, SN=5.

    p_RLC_Rec.UM_TX_Next := 5;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 8" siclog@

    //Wait for 200 ms to ensure that t-Reassembly for the UMD PDU#5 expires.

    f_Delay(v_T_ReassemblyMax);

    // @siclog "Step 9" siclog@

    //The SS transmits UMD PDU#6 containing last segment of RLC SDU#3, SN=5.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU3, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 10" siclog@

    //Check: For 1 sec after step 9, does the UE transmit RLC SDU#3?

    t_WatchDog.start(1.24); // as step 9 is scheduled: 1.0 sec + 240 msec

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");}

      [] t_WatchDog.timeout {}

    }

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment of SDU#4

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    // @siclog "Step 11" siclog@

    //The SS transmits UMD PDU#7 containing first segment of RLC SDU#4, SN=8.

    p_RLC_Rec.UM_TX_Next := 8;

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 12" siclog@

    //The SS transmits UMD PDU#8 containing last segment of RLC SDU#4, SN=8.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_SDUsize - v_RLC_Len1stSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 13" siclog@

    //Check: Does the UE transmit RLC SDU#4?

p_RLC_Rec.NR_RLC_DataList[3] := '8000021BD6C8E14FDE630F81'O;
    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");

  }

Change 4.2

	Function name
	f_TC_7_1_2_2_5_ENDC_NR

	Reason for change
	1. Out of delivery is not configured for PDCP in current TTCN, this will lead to UE not delivering the received RLC SDU#5 to test loop function and not looping back.
2. According to 38.523-1 Table 7.1.2.2.5.3.1-1, t-Reassembly should be 200ms which is not in current TTCN.

	Summary of change
	1. Change outOfOrderDelivery  to true for PDCP config.
2. Change v_T_Reassembly to 200ms.


	TTCN module
	RLC_ENDC_NR.ttcn

	MCC160 Comment
	


Before change

   function f_TC_7_1_2_2_5_ENDC_NR() runs on ENDC_NR_PTC

  { // UM RLC / Receive window operation and t-Reassembly expiry

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var SN_FieldLengthUM v_DL_Len;

    var SN_FieldLengthUM v_UL_Len;

    var T_Reassembly v_T_Reassembly;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    //Set null cipher Algorithm

    f_NR_AS_CipheringAlgorithm_Set(nea0); // @sic R5s190254 ch. 1.1 sic@

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    if (pc_um_WithShortSN) {

      v_DL_Len := size6;

      v_UL_Len := size6;

      v_RLC_Rec.UM_SN_Size := 6;

    } else {

      v_DL_Len := size12;

      v_UL_Len := size12;

      v_RLC_Rec.UM_SN_Size := 12;

    }

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_UM_Test(tsc_NR_DRB2, v_DL_Len);

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config); // PDCP default is 18 bits from 38.508

    v_T_Reassembly := f_NR_DL_UM_T_Reassembly(f_NR_CellInfo_GetIsFR1(nr_Cell1));
    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, v_DL_Len, v_UL_Len, v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

    f_NR_ENDC_ConfigAM_UM(nr_Cell1, 0, 1, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);

    // Synchronize with the EUTRA PTC

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Start TestBody"); // @sic R5s190163 sic@

    f_Delay (1.0);          //@sic R5s190254 sic@

    f_NR_TestBody_Set(true);

    f_TC_7_1_2_2_5_NR_TestBody(v_RLC_Rec, tsc_NR_DRB2);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(false);

    //Wait from coordination AFTER the connection is released in E-UTRA

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

    //Postamble

    f_NR_ReleaseAllCells();

  }

After change

    function f_TC_7_1_2_2_5_ENDC_NR() runs on ENDC_NR_PTC

  { // UM RLC / Receive window operation and t-Reassembly expiry

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var SN_FieldLengthUM v_DL_Len;

    var SN_FieldLengthUM v_UL_Len;

    var T_Reassembly v_T_Reassembly;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    //Set null cipher Algorithm

    f_NR_AS_CipheringAlgorithm_Set(nea0); // @sic R5s190254 ch. 1.1 sic@

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    if (pc_um_WithShortSN) {

      v_DL_Len := size6;

      v_UL_Len := size6;

      v_RLC_Rec.UM_SN_Size := 6;

    } else {

      v_DL_Len := size12;

      v_UL_Len := size12;

      v_RLC_Rec.UM_SN_Size := 12;

    }

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_UM_Test(tsc_NR_DRB2, v_DL_Len);

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config_OutOfOrderDelivery); // PDCP default is 18 bits from 38.508

v_T_Reassembly :=ms200; 

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, v_DL_Len, v_UL_Len, v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

    f_NR_ENDC_ConfigAM_UM(nr_Cell1, 0, 1, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);

    // Synchronize with the EUTRA PTC

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Start TestBody"); // @sic R5s190163 sic@

    f_Delay (1.0);          //@sic R5s190254 sic@

    f_NR_TestBody_Set(true);

    f_TC_7_1_2_2_5_NR_TestBody(v_RLC_Rec, tsc_NR_DRB2);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(false);

    //Wait from coordination AFTER the connection is released in E-UTRA

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

    //Postamble

    f_NR_ReleaseAllCells();

  }

Change 4.3

	template name
	cs_PDCP_Config_OutOfOrderDelivery,

cs_PDCP_Config_DRB_OutOfOrderDelivery

	Reason for change
	templates for PDCP Config with OutOfOrderDelivery set to true is missing in the current TTCN.


	Summary of change
	Add templates for PDCP Config with OutOfOrderDelivery set to true for TC_7_1_2_2_5.

	TTCN module
	RLC_ENDC_NR.ttcn

	MCC160 Comment
	


New template
  template (value) PDCP_Config cs_PDCP_Config_OutOfOrderDelivery(NR_PDCP_DiscardTimer_Type p_DiscardTimer := infinity_,





                                                       NR_PDCP_SN_UL_Type p_SN_UL := len18bits,





                                                       NR_PDCP_SN_DL_Type p_SN_DL := len18bits,





                                                       template (omit) NR_PDCP_T_Reordering_Type p_T_Reordering := omit) :=

  { 

    drb             := cs_PDCP_Config_DRB_OutOfOrderDelivery(p_DiscardTimer, p_SN_UL, p_SN_DL),

    moreThanOneRLC  := omit,

    t_Reordering    := p_T_Reordering,

    cipheringDisabled := omit          

   } // RLC_AM_Templates

 template (value) PDCP_Config.drb cs_PDCP_Config_DRB_OutOfOrderDelivery(NR_PDCP_DiscardTimer_Type p_DiscardTimer := infinity_,





                                                               NR_PDCP_SN_UL_Type p_SN_UL := len18bits,





                                                               NR_PDCP_SN_DL_Type p_SN_DL := len18bits) :=

  { 

    discardTimer        := p_DiscardTimer,

    pdcp_SN_SizeUL      := p_SN_UL,

    pdcp_SN_SizeDL      := p_SN_DL,

    headerCompression   := cs_NR_PDCP_HeaderCompressionNotUsed,

    integrityProtection := omit,

    statusReportRequired:= true_,

    outOfOrderDelivery  := true_

  };

Change 4.4

	Const name
	tsc_NR_UM_Window_Size_6

	Reason for change
	According to TS 38.322 cl.7.2, UM_Window_Size should be 32.

	Summary of change
	Change UM_Window_Size from 64 to 32

	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	


Before change

   const integer tsc_NR_UM_Window_Size_6 := 64;  
After change

   const integer tsc_NR_UM_Window_Size_6 := 32; 
5. Execution Log Files

5.1 HiSilicon Balong 5000 & MediaTek MT6297
The HiSilicon Balong 5000 and MediaTek MT6297 UE passed this test case on Starpoint SP9500 Test System on EN-DC 3A-n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
\Hisilicon\ TC_7_1_2_2_5_ENDC_PASS.spm
\MTK\ TC_7_1_2_2_5_ENDC_PASS.spm
PIXCIT settings used:

 \ ENDC_Testsuite.sppar

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

